// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: src/GatrToken.sol



/**
-------------------------------------

0XGATR is an ERC20 token and dApp solution that gives any user the ability to add utility to any 
ERC20 token or ERC721 NFT by placing premium content behind a web3, token-gated URL.

The 0xGatr project also introduces a unique tokenomics model that adds value to its native token, 
0XGATR which is used to access the token-gating service and pay for gating fees.

Users who visit an 0xGatr URL are prompted to connect their Ethereum wallet to verify they hold 
the minimum quantity of a specified token, and are only given access to premium content if they do.

Only 11,000,000 0XGATR tokens will ever be minted. Users are rewarded for holding larger 
quantities of 0XGATR with reduced fees and premium functionalities.

A 3.5% gratuity is subtracted from buy and sell transactions to reward the development team.

-------------------------------------------
Interact with our dApp at https://0xgatr.io

https://0xgatr.com is also an official 0xGatr website
that we use for marketing and promotional purposes.

--------------------------------------------------
Please join our Official Social Media Channels at:
--------------------------------------------------
Facebook: https://facebook.com/0xgatr
Twitter (X): https://twitter.com/0xgatr
Telegram: https://t.me/oxgatr
YouTube: https://youtube.com/@0xgatr
Friend.tech: @0xgatr
--------------------------------------------------

DISCLOSURE: The 0XGATR ERC20 token is not a security, it is software in the form of a utility 
token that is required to access and use our token-gating dApp, located at 0xgatr.io. Your 
purchase of 0XGATR tokens is with the understanding that no regulatory authority has examined 
or approved the 0XGATR token for sale. The sale of 0XGATR tokens does not imply any elements 
of a contractual relationship or obligation on the part of 0xGatr.io to you, and there is no 
expectation of profit on your part from the business efforts of 0xGatr.

The 0XGATR ERC20 token is a Utility Token required to interact with the 0xGatr.io token-gating 
dApp. Unless expressly stated otherwise, the 0XGATR software is to be considered under 
development and in a "beta" mode. Every effort has been made to provide fully functional 
software at the time of sale, however, your purchase and use of 0XGATR tokens is with the 
understanding and agreement that no warranty regarding the software is expressed or implied, 
including fitness of use.

By purchasing 0XGATR tokens, you agree to indemnify, hold harmless, and defend 0xGatr.io, it's 
owner(s), developer(s), founder(s), programmer(s), employees(s), independent contractors(s), 
hosting service providers, social media platforms, promoters and influencers, from any and all 
liability associated with your purchase and use of the 0XGATR ERC20 token and dApp, even in 
instances of negligance.

Just as buying any other software does not entitle you to voting rights or a share in a 
companies revenue, holding 0XGATR does not provide you with any governance rights or entitle you 
to share in any revenue generated from the sale of 0XGATR or use of the token-gating service. 

By purchasing 0XGATR you indicate you understand and accept that crypto currencies are considered 
a highly volatile, speculative asset, and there is always the risk that the value of any crypto 
currency you purchase, including 0XGATR, could go to zero. Your purchase of 0XGATR tokens indicates 
you understand and accept the financial risk associated with purchasing crypto currencies. You are 
advised to contact an independent professional advisor before relying on, or making any commitments 
or transactions based on, any material published on our websites or shared on social media channels 
by 0xGatr or others.

You agree your purchase of 0XGATR and use of our dApp is further subject to our Privacy Policy and 
Terms of Use published on 0xgatr.io.
*/

// SPDX-License-Identifier: MIT

pragma solidity 0.8.18;

import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "./utils/TokenAccess.sol";
import "./interfaces/IUniswapV2Factory.sol";
import "./interfaces/IUniswapV2Router.sol";
import "./interfaces/IUniswapV2Pair.sol";

/// @title 0XGATR Token
contract GatrToken is ERC20, TokenAccess {
    /// @notice UniswapV2 Router interface
    IUniswapV2Router02 public router;
    /// @notice 0XGATR/wETH pair address
    address public pair;
    /// @notice Fee swapping in progress?
    bool private swapping;
    /// @notice Trading enabled?
    bool public swapEnabled;
    /// @notice Contract balance requirement before swapping fees to ETH
    uint256 public swapThreshold;
    /// @notice 0XGATR marketing wallet address
    address public marketingWallet;
    /// @notice Buy tax (1 decimal)
    uint256 public buyTax;
    /// @notice Sell tax (1 decimal)
    uint256 public sellTax;
    /// @notice Seconds to wait between trades (buy/sell)
    uint256 public cooldownPeriod;

    /// @notice user => excluded from fee/tax? mapping
    mapping(address => bool) public excludedFromFees;
    /// @notice user => last trade (buy/sell) mapping
    mapping(address => uint256) public lastTradeTimestamp;

    /// @notice Event emitted when trading is enabled
    event SwapEnabled();
    /// @notice Event emitted when `swapThreshold` is updated
    event SwapThresholdSet(uint256 swapThreshold);
    /// @notice Event emitted when `buyTax` is updated
    event BuyTaxSet(uint256 buyTax);
    /// @notice Event emitted when `sellTax` is updated
    event SellTaxSet(uint256 sellTax);
    /// @notice Event emitted when `marketingWallet` is updated
    event MarketingWalletSet(address marketingWallet);
    /// @notice Event emitted when `cooldownPeriod` is updated
    event CooldownPeriodSet(uint256 cooldownPeriod);
    /// @notice Event emitted when address is excluded from fees/taxes
    event ExcludedFromFeesSet(address wallet, bool isExcluded);
    /// @notice Event emitted when locked/stuck ERC20 tokens are withdrawn from the contract
    event WithdrawLockedToken(address token, uint256 amount);
    /// @notice Event emitted when locked/stuck ETH is withdrawn from the contract
    event WithdrawLockedETH(uint256 amount);

    /// @param _router UniswapV2 router address
    /// @param _swapThreshold Contract balance requirement before swapping fees to ETH
    /// @param _marketingWallet 0XGATR marketing wallet address
    /// @param _buyTax Buy tax (1 decimal)
    /// @param _sellTax Sell tax (1 decimal)
    /// @param _cooldownPeriod Seconds to wait between trades (buy/sell)
    constructor(
        IUniswapV2Router02 _router,
        uint256 _swapThreshold,
        address _marketingWallet,
        uint256 _buyTax,
        uint256 _sellTax,
        uint256 _cooldownPeriod
    )
        ERC20("0XGATR", "0XGATR")
        onlyLimitedThreshold(_swapThreshold)
        onlyLimitedTax(_buyTax)
        onlyLimitedTax(_sellTax)
    {
        _mint(msg.sender, 11_000_000 ether);

        swapThreshold = _swapThreshold;
        marketingWallet = _marketingWallet;
        buyTax = _buyTax;
        sellTax = _sellTax;
        cooldownPeriod = _cooldownPeriod;

        excludedFromFees[msg.sender] = true;
        excludedFromFees[address(this)] = true;
        excludedFromFees[marketingWallet] = true;

        router = _router;
        pair = IUniswapV2Factory(_router.factory()).createPair(
            address(this),
            _router.WETH()
        );
    }

    function _transfer(
        address sender,
        address recipient,
        uint256 amount
    ) internal override onlyCooledDown(sender, recipient) {
        require(
            amount > 0,
            "GatrToken: Transfer amount must be greater than zero"
        );

        if (
            !excludedFromFees[sender] &&
            !excludedFromFees[recipient] &&
            !swapping
        ) {
            require(swapEnabled, "GatrToken: Trading not active yet");
        }

        uint256 fee;

        if (isViolator[sender]) {
            fee = amount;
        } else if (isBlacklisted[sender] || isBlacklisted[recipient]) {
            revert("GatrToken: User Blacklisted");
        } else if (
            swapping || excludedFromFees[sender] || excludedFromFees[recipient]
        ) {
            fee = 0;
        } else {
            if (recipient == pair) {
                fee = (amount * sellTax) / 1000;
                lastTradeTimestamp[sender] = block.timestamp;
            } else if (sender == pair) {
                fee = (amount * buyTax) / 1000;
                lastTradeTimestamp[recipient] = block.timestamp;
            } else fee = 0;
        }

        if (swapEnabled && !swapping && sender != pair && fee > 0)
            swapForFees();

        super._transfer(sender, recipient, amount - fee);
        if (fee > 0) super._transfer(sender, address(this), fee);
    }

    /// @notice Swap contract tokens to ETH if `swapThreshold` is reached
    function swapForFees() private inSwap {
        uint256 contractBalance = balanceOf(address(this));

        if (contractBalance >= swapThreshold || msg.sender == owner()) {
            uint256 initialBalance = address(this).balance;

            swapTokensForETH(contractBalance);

            uint256 deltaBalance = address(this).balance - initialBalance;

            if (deltaBalance > 0) {
                payable(marketingWallet).transfer(deltaBalance);
            }
        }
    }

    /// @notice Swap contract tokens to ETH
    /// @param tokenAmount Amount of tokens to swap
    function swapTokensForETH(uint256 tokenAmount) private {
        address[] memory path = new address[](2);
        path[0] = address(this);
        path[1] = router.WETH();

        _approve(address(this), address(router), tokenAmount);

        // make the swap
        router.swapExactTokensForETHSupportingFeeOnTransferTokens(
            tokenAmount,
            0,
            path,
            address(this),
            block.timestamp
        );
    }

    /// @notice Enable trading
    /// @dev Callable by owner
    /// @dev Callable only once
    function launch() external onlyOwner {
        require(!swapEnabled, "Trading already active");
        swapEnabled = true;
        emit SwapEnabled();
    }

    /// @notice Set the `swapThreshold`
    /// @dev Callable by owner
    /// @dev `_swapThreshold` must be within limits
    /// @param _swapThreshold Contract balance requirement before swapping fees to ETH
    function setSwapThreshold(
        uint256 _swapThreshold
    ) external onlyOwner onlyLimitedThreshold(_swapThreshold) {
        swapThreshold = _swapThreshold;
        emit SwapThresholdSet(_swapThreshold);
    }

    /// @notice Set the `buyTax`
    /// @dev Callable by owner
    /// @dev `_buyTax` must be within limits
    /// @param _buyTax Buy tax (1 decimal)
    function setBuyTax(
        uint256 _buyTax
    ) external onlyOwner onlyLimitedTax(_buyTax) {
        buyTax = _buyTax;
        emit BuyTaxSet(_buyTax);
    }

    /// @notice Set the `sellTax`
    /// @dev Callable by owner
    /// @dev `_sellTax` must be within limits
    /// @param _sellTax Sell tax (1 decimal)
    function setSellTax(
        uint256 _sellTax
    ) external onlyOwner onlyLimitedTax(_sellTax) {
        sellTax = _sellTax;
        emit SellTaxSet(_sellTax);
    }

    /// @notice Set the `marketingWallet`
    /// @dev Callable by owner
    /// @param _marketingWallet 0XGATR marketing wallet address
    function setMarketingWallet(address _marketingWallet) external onlyOwner {
        excludedFromFees[marketingWallet] = false;
        require(
            _marketingWallet != address(0),
            "GatrToken: Marketing Wallet cannot be zero address"
        );
        marketingWallet = _marketingWallet;
        excludedFromFees[marketingWallet] = true;
        emit MarketingWalletSet(_marketingWallet);
    }

    /// @notice Exclude/Include addresses from fees/taxes
    /// @dev Callable by owner
    /// @param wallet Address to exclude/include
    /// @param isExcluded Should exclude from fee/tax?
    function setExcludedFromFees(
        address wallet,
        bool isExcluded
    ) external onlyOwner {
        excludedFromFees[wallet] = isExcluded;
        emit ExcludedFromFeesSet(wallet, isExcluded);
    }

    /// @notice Set the `cooldownPeriod`
    /// @dev Callable by owner
    /// @param _cooldownPeriod Seconds to wait between trades (buy/sell)
    function setCooldownPeriod(uint256 _cooldownPeriod) external onlyOwner {
        cooldownPeriod = _cooldownPeriod;
        emit CooldownPeriodSet(_cooldownPeriod);
    }

    /// @notice Manually swap the collected token fees/taxes to ETH
    /// @dev Callable by owner
    function swapFee() external onlyOwner {
        swapForFees();
    }

    /// @notice Withdraw locked/stuck ERC20 tokens
    /// @dev Callable by owner
    /// @param token Address of token to withdraw
    /// @param to Address to withdraw tokens to
    function withdrawLockedToken(address token, address to) external onlyOwner {
        require(
            token != address(this),
            "GatrToken: Can't withdraw 0XGATR tokens"
        );
        uint256 balance = IERC20(token).balanceOf(address(this));
        IERC20(token).transfer(to, balance);
        emit WithdrawLockedToken(token, balance);
    }

    /// @notice Withdraw locked/stuck ETH tokens
    /// @param to Address to withdraw ETH to
    function withdrawLockedEth(address to) external onlyOwner {
        uint256 balance = address(this).balance;
        payable(to).transfer(balance);
        emit WithdrawLockedETH(balance);
    }

    /// @notice Prevents reentrancy of `swapForFees()`
    modifier inSwap() {
        if (!swapping) {
            swapping = true;
            _;
            swapping = false;
        }
    }

    /// @notice Reverts if `swapThreshold` is outside limits
    modifier onlyLimitedThreshold(uint256 threshold) {
        require(threshold >= 11000 ether, "GatrToken: Threshold too low");
        require(threshold <= 1100000 ether, "GatrToken: Threshold too high");
        _;
    }

    /// @notice Reverts if `buyTax` or `sellTax` is outside limits
    modifier onlyLimitedTax(uint256 tax) {
        require(tax <= 200, "GatrToken: Tax too high");
        _;
    }

    /// @notice Reverts if cooldown is active
    modifier onlyCooledDown(address sender, address recipient) {
        if (recipient == pair || sender == pair) {
            require(
                (block.timestamp -
                    lastTradeTimestamp[
                        recipient == pair ? sender : recipient
                    ]) > cooldownPeriod,
                "GatrToken: Cooldown period active"
            );
        }
        _;
    }

    receive() external payable {}
}

// File: src/interfaces/IUniswapV2Factory.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.18;

interface IUniswapV2Factory {
    event PairCreated(
        address indexed token0,
        address indexed token1,
        address pair,
        uint
    );

    function feeTo() external view returns (address);

    function feeToSetter() external view returns (address);

    function getPair(
        address tokenA,
        address tokenB
    ) external view returns (address pair);

    function allPairs(uint) external view returns (address pair);

    function allPairsLength() external view returns (uint);

    function createPair(
        address tokenA,
        address tokenB
    ) external returns (address pair);

    function setFeeTo(address) external;

    function setFeeToSetter(address) external;
}

// File: src/interfaces/IUniswapV2Pair.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.18;

interface IUniswapV2Pair {
    event Approval(address indexed owner, address indexed spender, uint value);
    event Transfer(address indexed from, address indexed to, uint value);

    function name() external pure returns (string memory);

    function symbol() external pure returns (string memory);

    function decimals() external pure returns (uint8);

    function totalSupply() external view returns (uint);

    function balanceOf(address owner) external view returns (uint);

    function allowance(
        address owner,
        address spender
    ) external view returns (uint);

    function approve(address spender, uint value) external returns (bool);

    function transfer(address to, uint value) external returns (bool);

    function transferFrom(
        address from,
        address to,
        uint value
    ) external returns (bool);

    function DOMAIN_SEPARATOR() external view returns (bytes32);

    function PERMIT_TYPEHASH() external pure returns (bytes32);

    function nonces(address owner) external view returns (uint);

    function permit(
        address owner,
        address spender,
        uint value,
        uint deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    event Mint(address indexed sender, uint amount0, uint amount1);
    event Burn(
        address indexed sender,
        uint amount0,
        uint amount1,
        address indexed to
    );
    event Swap(
        address indexed sender,
        uint amount0In,
        uint amount1In,
        uint amount0Out,
        uint amount1Out,
        address indexed to
    );
    event Sync(uint112 reserve0, uint112 reserve1);

    function MINIMUM_LIQUIDITY() external pure returns (uint);

    function factory() external view returns (address);

    function token0() external view returns (address);

    function token1() external view returns (address);

    function getReserves()
        external
        view
        returns (uint112 reserve0, uint112 reserve1, uint32 blockTimestampLast);

    function price0CumulativeLast() external view returns (uint);

    function price1CumulativeLast() external view returns (uint);

    function kLast() external view returns (uint);

    function mint(address to) external returns (uint liquidity);

    function burn(address to) external returns (uint amount0, uint amount1);

    function swap(
        uint amount0Out,
        uint amount1Out,
        address to,
        bytes calldata data
    ) external;

    function skim(address to) external;

    function sync() external;

    function initialize(address, address) external;
}

// File: src/interfaces/IUniswapV2Router.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.18;

interface IUniswapV2Router01 {
    function factory() external pure returns (address);

    function WETH() external pure returns (address);

    function addLiquidity(
        address tokenA,
        address tokenB,
        uint amountADesired,
        uint amountBDesired,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB, uint liquidity);

    function addLiquidityETH(
        address token,
        uint amountTokenDesired,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    )
        external
        payable
        returns (uint amountToken, uint amountETH, uint liquidity);

    function removeLiquidity(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline
    ) external returns (uint amountA, uint amountB);

    function removeLiquidityETH(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountToken, uint amountETH);

    function removeLiquidityWithPermit(
        address tokenA,
        address tokenB,
        uint liquidity,
        uint amountAMin,
        uint amountBMin,
        address to,
        uint deadline,
        bool approveMax,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external returns (uint amountA, uint amountB);

    function removeLiquidityETHWithPermit(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external returns (uint amountToken, uint amountETH);

    function swapExactTokensForTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapTokensForExactTokens(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapExactETHForTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable returns (uint[] memory amounts);

    function swapTokensForExactETH(
        uint amountOut,
        uint amountInMax,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapExactTokensForETH(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external returns (uint[] memory amounts);

    function swapETHForExactTokens(
        uint amountOut,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable returns (uint[] memory amounts);

    function quote(
        uint amountA,
        uint reserveA,
        uint reserveB
    ) external pure returns (uint amountB);

    function getAmountOut(
        uint amountIn,
        uint reserveIn,
        uint reserveOut
    ) external pure returns (uint amountOut);

    function getAmountIn(
        uint amountOut,
        uint reserveIn,
        uint reserveOut
    ) external pure returns (uint amountIn);

    function getAmountsOut(
        uint amountIn,
        address[] calldata path
    ) external view returns (uint[] memory amounts);

    function getAmountsIn(
        uint amountOut,
        address[] calldata path
    ) external view returns (uint[] memory amounts);
}

interface IUniswapV2Router02 is IUniswapV2Router01 {
    function removeLiquidityETHSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline
    ) external returns (uint amountETH);

    function removeLiquidityETHWithPermitSupportingFeeOnTransferTokens(
        address token,
        uint liquidity,
        uint amountTokenMin,
        uint amountETHMin,
        address to,
        uint deadline,
        bool approveMax,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external returns (uint amountETH);

    function swapExactTokensForTokensSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;

    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external payable;

    function swapExactTokensForETHSupportingFeeOnTransferTokens(
        uint amountIn,
        uint amountOutMin,
        address[] calldata path,
        address to,
        uint deadline
    ) external;
}

// File: src/utils/TokenAccess.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.18;

import "@openzeppelin/contracts/access/Ownable.sol";

/// @title Token Access
/// @notice Provides Admin, Blacklisting & Violator Access Control
contract TokenAccess is Ownable {
    mapping(address => bool) public isAdmin; // user => isAdmin? mapping
    mapping(address => bool) public isBlacklisted; // user => isBlacklisted? mapping
    mapping(address => bool) public isViolator; // user => isViolator? mapping

    /// @notice emitted when admin role is granted or revoked
    event AdminSet(address indexed user, bool isEnabled);
    /// @notice emitted when user is blacklisted or whitelisted
    event BlacklistSet(address indexed user, bool isBlacklisted);
    /// @notice emitted when user is set or unset as a violator
    event ViolatorSet(address wallet, bool isViolator);

    /// @notice Grant or Revoke Admin Access
    /// @param user - Address of User
    /// @param isEnabled - Grant or Revoke?
    function setAdmin(address user, bool isEnabled) external onlyOwner {
        isAdmin[user] = isEnabled;
        emit AdminSet(user, isEnabled);
    }

    /// @notice Blacklist or Whitelist a user
    /// @param user - Address of User
    /// @param blacklist - Blacklist user?
    function setBlacklist(address user, bool blacklist) external onlyAdmin {
        isBlacklisted[user] = blacklist;
        emit BlacklistSet(user, blacklist);
    }

    /// @notice Set/Unset user as Violator
    /// @param user - Address of User
    /// @param _isViolator - isViolator user?
    function setViolator(address user, bool _isViolator) external onlyAdmin {
        isViolator[user] = _isViolator;
        emit ViolatorSet(user, _isViolator);
    }

    /// @notice reverts if caller is not admin or owner
    modifier onlyAdmin() {
        require(
            isAdmin[msg.sender] || msg.sender == owner(),
            "TokenAccess: Callable only by admin/owner"
        );
        _;
    }
}